

## 22. Words

### *General Description*

This and the next program form a pair which should be considered together. They read and set up a 'language file' of a limited amount of words. There are no educational notes for the program as it was not developed for use in schools but as an afterthought by a programmer teaching himself Welsh. It is included for the exciting opportunities it offers in allowing young people to explore the vocabulary of a language they are studying. So many programs of this ilk are repetitive testing and this is the first I have come across that is remotely 'developmental'.

The program is virtually self-documenting and so the line descriptions can be kept short. However the printed program is long and you are welcome, after reading it, to shorten variable names etc. when typing it in. Be sure to type in the PRINT statements precisely as written. The odd words joined together are deliberate and come at the end of screen lines, and so an automatic CR/LF is generated. I do not generally approve of this practice, but in this case it aids the general readability of the program.

### *Detailed Description*

**Lines 10-290** The main structure: the auto-repeat has been deliberately turned off at line 100.

**300-680** All are general purpose routines. Remember the warning about the PRINT statement and you could alter the programme by restructuring it around the validation routines suggested elsewhere in the book.

**900-1020** There is an easier way of doing this, however this

routine works.

**1030-1420** Self-evident: at line 1210 ' UNTILNO' is sufficient, but as documented is clearer. In line 1410 FNCONV is called, I suggest that, for clarity and shortness, you may wish to use the FNconvert included elsewhere in the book.

**1430-1530** Again, this works but there is a more elegant way of doing it, see comment above.

**1540-1860** This is in some ways the heart of the program. It is this routine that puts a usable maximum of words that can be defined as around 250, as the search will then take too long to be productive.

**1870-2480** Well self-documented, the delay routine is as in the User Guide.

**2490-2930** At this point you should be aware of the known ' bugs in the BBC saving of Cassette Files. BEEBUG magazine have these fully documented, but if you do not use the ' fix you may come across files that you cannot read later. However I have tested this program and not been able to generate the ' saving' error. Note £ prints instead of # in the Save and Load routines, as noted before.

**2940-4000** If you are at all familiar with programming, you will find nothing clever here. All the functional routines described in the main structure reside here. The author of the program tells me that he has not supplied a sort routine, for with the limited number of words he was using in his dictionary he could see no need for one. If you do, then write it, can call it from the main structure. However be careful when altering the MENU display.

## *Second Program*

This program produces a language tape for the former program. To alter the language change the DATA at line 790 and 800, as well as the words themselves otherwise the program assumes 50 words in the language.

Once you have typed in the program, you would need to save it and rewrite lines 750-1340 as you needed them.

## Program Listing

```

10 REM
20 REM =====
30 REM      WORDS
40 REM
50 REM      BY CW FEB 1982
60 REM .....
70 REM
80 ON ERROR GOTO 4050
90 *KEY10 OLD|MRUN|M
100 *FX11,0
110 DIM WORD$(500,1),LANG$(2)
120 MAXWD%=1
130 MODE 7
140 PROC_NAME
150 PROC_INSTR
160 REPEAT
170   PROC_MENU
180   ON SELECT% GOTO 190,200,210,220,230,240,250,260
190   PROC_INSERT: GOTO 260
200   PROC_AMEND: GOTO 260
210   PROC_DELETE: GOTO 260
220   PROC_LOOK_UP: GOTO 260
230   PROC_LIST: GOTO 260
240   PROC_SAVE: GOTO 260
250   PROC_LOAD: GOTO 260
260 UNTIL SELECT%=8
270 CLS
280 *FX12,0
290 END
300 REM =====
310 REM      PROC_NAME
320 REM .....
330 REM
340 DEF PROC_NAME
350 CLS
360 PRINT TAB(14,8);CHR$131;CHR$141;"W O R D S";TAB(14,9);
CHR$131;CHR$141;"W O R D S"
370 PRINT TAB(11,14);CHR$134;"By C.W. Feb 1983"
380 PROC_DELAY(200)
390 ENDPROC
400 REM =====
410 REM      PROC_INSTR
420 REM .....
430 REM
440 DEF PROC_INSTR
450 CLS:PROC_TITLE
460 PRINT" This is a vocabulary program. You typewords i
n one language and the programreplies with the word i
n the secondlanguage."
470 PRINT" The program allows you to insert newwords,
amend existing words and deleteexisting words."
480 PRINT" Existing vocabularies can be loadedfrom ta
pe, and new vocabularies can besaved to tape."
490 PRINT" You can return to the Main Menu at anytime b
y just pressing ";CHR$130;"RETURN";CHR$135;" to anyquestio
n."
500 PROC_NEXT
510 ENDPROC
520 REM =====
530 REM      PROC_TITLE
540 REM .....
550 REM
560 DEF PROC_TITLE
570 PRINT TAB(16);CHR$131;CHR$141;"WORDS"
580 PRINT TAB(16);CHR$131;CHR$141;"WORDS"

```

```

590 ENDPROC
600 REM =====
610 REM   PROC_NEXT
620 REM .....
630 REM
640 DEF PROC_NEXT
650 *FX15,0
660 PRINT TAB(0,23);CHR$129;"Press any key to continue";
670 NX$=GET$
680 ENDPROC
690 REM =====
700 REM   PROC_MENU
710 REM .....
720 REM
730 DEF PROC_MENU
740 CLS: PROC_TITLE
750 PRINT'CHR$129;CHR$141;"M E N U"
760 PRINT CHR$129;CHR$141;"M E N U"
770 PRINT'CHR$134;"1. Insert a word"
780 PRINT CHR$134;"2. Amend a word"
790 PRINT CHR$134;"3. Delete a word"
800 PRINT CHR$134;"4. Look up a word"
810 PRINT CHR$134;"5. List the vocabulary"
820 PRINT CHR$134;"6. SAVE a vocabulary"
830 PRINT CHR$134;"7. LOAD a vocabulary"
840 PRINT CHR$134;"8. End"
850 *FX15,0
860 PRINT''CHR$130;CHR$136;"TYPE YOUR SELECTION (1,-8)";
CHR$137;CHR$133;"?";:SELECTION$=GET$
870 SELECT%=FNNUM(SELECTION$,1,8,0)
880 IF SELECT%=FALSE THEN GOTO 740
890 ENDPROC
900 REM =====
910 REM   FNNUM - CHECK NUMERIC
920 REM .....
930 REM
940 DEF FNNUM(A$,B%,C%,D%)
950 LOCAL I$,B$
960 B$=""
970 FOR I%=1 TO LEN(A$)
980   IF MID$(A$,I%,1)<"0" OR MID$(A$,I%,1)>"9" THEN B$=ST
R$(D%): I%=LEN(A$)
990 NEXT I%
1000 IF B$<>STR$(D%) THEN B$=A$
1010 IF VAL(B$)<B% OR VAL(B$)>C% THEN B$=STR$(D%)
1020 B$=VAL(B$)
1030 REM =====
1040 REM   PROC_INSERT
1050 REM .....
1060 REM
1070 DEF PROC_INSERT
1080 IF LANG$(1)="" OR LANG$(2)="" THEN PROC_LANGUAGE
1090 IF LANG$(1)="" THEN ENDPROC
1100 CLS: PROC_TITLE
1110 PRINT'CHR$134;"INSERT A WORD"
1120 PRINT" You will be asked to specify a word in the fir
st language:-"
1130 PRINT CHR$130;LANG$(1)
1140 PRINT" Then a word in the second language:-"
1150 PRINT CHR$133;LANG$(2)
1160 PRINT" The program stores the pair of words for you
so you can recall them later."
1170 PRINT" You will then be asked if you want to add ano
ther pair. This will continue until you reply 'N'."
1180 PROC_NEXT
1190 REPEAT
1200   PROC_INSERT_WORD
1210   PROC_YORN
1220 UNTIL NO=TRUE

```

```

1230 ENDPROC
1240 REM =====
1250 REM     PROC_LANGUAGE
1260 REM .....
1270 REM
1280 DEF PROC_LANGUAGE

1290 CLS:PROC_TITLE
1300 PRINT'CHR$134;"INSERT A WORD"
1310 PRINT'" First you must tell the program thenames o
f the two languages you are using"
1320 PRINT'" To get back to the main menu pressRETURN
in reply to any question"
1330 PRINT'" Type the name of your first language"
1340 INPUT LANG$(1)
1350 PRINT'" Now type your second language"
1360 INPUT LANG$(2)
1370 IF LANG$(1)="" OR LANG$(2)="" THEN LANG$(1)="": LANG$(
2)="":ENDPROC
1380 PRINT '"Are the names correct?";
1390 PROC_YORN
1400 IF NO THEN GOTO 1290
1410 LANG$(1)=FNCONV(LANG$(1)):LANG$(2)=FNCONV(LANG$(2))
1420 ENDPROC
1430 REM =====
1440 REM     PROC_Y OR N
1450 REM .....

1460 REM
1470 DEF PROC_YORN
1480 LOCAL Z$
1490 Z$=GET$
1500 IF Z$="Y" OR Z$="y" THEN YES=TRUE: NO=FALSE
1510 IF Z$="N" OR Z$="n" THEN NO=TRUE: YES=FALSE
1520 IF Z$<>"Y" AND Z$<>"N" AND Z$<>"y" AND Z$<>"n" THEN GO
TO 1490
1530 ENDPROC
1540 REM =====
1550 REM     PROC_INSERT_WORD
1560 REM .....
1570 REM
1580 DEF PROC_INSERT_WORD
1590 CLS:PROC_TITLE
1600 DUP=FALSE
1610 PRINT'CHR$134"INSERT A WORD"
1620 PRINT'" To exit to the Main Menu press RETURNin repl
y to any question."
1630 PRINT'" Type in the word in ";LANG$(1); ":-"
1640 INPUT WORD1$
1650 IF WORD1$="" THEN GOTO 1850
1660 IF MAXWD%=1 GOTO 1710
1670 FOR I%=1 TO MAXWD%-1
1680 IF WORD1$=WORD$(I%,0) THEN PRINT CHR$129;"Word alrea
dy defined": I%=MAXWD%: DUP=TRUE
1690 NEXT I%
1700 IF DUP GOTO 1590
1710 PRINT'" Type the word in ";LANG$(2); ":-"
1720 INPUT WORD2$
1730 IF MAXWD%=1 THEN GOTO1820
1740 IF WORD2$="" THEN GOTO1850
1750 FOR I%=1 TO MAXWD%
1760 IF WORD2$=WORD$(I%,1) THEN PRINT CHR$129;"Word alrea
dy defined": I%=MAXWD%: DUP=TRUE
1770 NEXT I%
1780 IF DUP GOTO 1850
1790 PRINT'"Are these words correct (Y/N)";
1800 PROC_YORN
1810 IF NO THEN GOTO 1590
1820 WORD$(MAXWD%,0)=FNCONV(WORD1$)

```

```

1830 WORD$(MAXWD%,1)=FNCONV(WORD2$)
1840 MAXWD%=MAXWD%+1
1850 PRINT TAB(0,23);CHR$129;"Another word (Y/N)? ";
1860 ENDPROC
1870 REM =====
1880 REM      CONVERT TO UPPER CASE
1890 REM .....
1900 DEF FNCONV(A$)
1910 LOCAL B$,Y%
1920 B$=""
1930 FOR Y%=1 TO LEN(A$)
1940   IF ASC(MID$(A$,Y%,1))>90 THEN B$=B$+CHR$(ASC(MID$(A$,Y%,1))-32) ELSE B$=B$+MID$(A$,Y%,1)
1950 NEXT Y%
1960 =B$
1970 REM =====
1980 REM      PROC_AMEND
1990 REM .....
2000 REM
2010 DEF PROC_AMEND
2020 CLS:PROC_TITLE
2030 IF MAXWD%=1 THEN PRINT'''CHR$129;"You must enter some words before you": PRINT CHR$129;"Can amend them!!!!!!":PROC_NEXT:ENDPROC
2040 PRINT'CHR$134;"AMEND A WORD"
2050 PRINT'" You will be asked to give the language of the word you wish to amend and then the word. The old word in the other language will be displayed and you will be asked for the new word.'
2060 PRINT'" You will then be asked if you want to amend another word. This will continue until you reply No.'
2070 PROC_NEXT
2080 REPEAT
2090   PROC_AMEND_WORD
2100   PROC_YORN
2110 UNTIL NO=TRUE
2120 ENDPROC
2130 REM =====
2140 REM      PROC_AMEND_WORD
2150 REM .....
2160 REM
2170 DEF PROC_AMEND_WORD
2180 CLS: PROC_TITLE
2190 PRINT'CHR$134;"AMEND A WORD"
2200 PRINT'" To exit to the Main Menu press RETURN in reply to any question.'
2210 PRINT'"Type the language":INPUT LANG$
2220 IF LANG$="" THEN GOTO 2380
2230 LANG$=FNCONV(LANG$)
2240 IF LANG$<>LANG$(1) AND LANG$<>LANG$(2) THEN PRINT'CHR$129;"Language unknown!":GOTO 2380
2250 IF LANG$=LANG$(1) THEN LANG%=0 ELSE LANG%=1
2260 PRINT'"Type the word to amend in ";LANG$(LANG%+1): INPUT WORD$
2270 WORD$=FNCONV(WORD$)
2280 IF WORD$="" THEN GOTO 2380
2290 DUP=FALSE
2300 FOR I%=1 TO MAXWD%-1
2310   IF WORD$=WORD$(I%,LANG%) THEN DUP=TRUE: J%=I%:I%=MAXWD%-1
2320 NEXT I%
2330 IF NOT DUP THEN PRINT'CHR$129;"No such word!!!!!!": GOTO 2380
2340 PRINT'"Type new word in ";LANG$((LANG%+1) MOD 2+1): INPUT WORD$
2350 IF WORD$="" THEN GOTO 2380
2360 WORD$=FNCONV(WORD$)
2370 WORD$(J%,(LANG%+1) MOD 2)=WORD$
2380 PRINT TAB(0,23);CHR$129;"Another word to amend (Y/N)";

```

```

2390 ENDPROC
2400 REM =====
2410 REM   PROC_DELAY
2420 REM .....
2430 REM
2440 DEF PROC_DELAY(X%)
2450 NOW=TIME
2460 REPEAT
2470 UNTIL TIME=NOW+X%
2480 ENDPROC
2490 REM =====
2500 REM   PROC_SAVE
2510 REM .....
2520 REM
2530 DEF PROC_SAVE
2540 CLS:PROC_TITLE
2550 IF MAXWD%=1 THEN PRINT'''CHR$129;"You must enter some
words before you":PRINT CHR$129;"can save them!!!!!!":GOT
O 2700
2560 PRINT'CHR$134;"SAVE A VOCABULARY"
2570 PRINT'"   To exit to the Main Menu press   RETURNin repl
y to the name of the file"
2580 PRINT'"   Save the vocabulary you   have   createdor amen
ded to tape."
2590 PRINT'"   Put a tape in your cassette recorder."
2600 PRINT'"   What is the name of the file":INPUT FILE$

2610 IF FILE$="" THEN GOTO 2710
2620 PRINT'"Press any key to continue":Z$=GET$
2630 FILE=OPENOUT(FILE$)
2640 PRINT# FILE,LANG$(1),LANG$(2),MAXWD%
2650 FOR I%=1 TO MAXWD%-1
2660   PRINT# FILE,WORD$(I%,0),WORD$(I%,1)
2670 NEXT I%

2680 CLOSE#FILE
2690 PRINT'FILE$;" Saved"
2700 PROC_NEXT
2710 ENDPROC
2720 REM =====
2730 REM   PROC_LOAD

2740 REM .....
2750 REM
2760 DEF PROC_LOAD
2770 CLS:PROC_TITLE
2780 PRINT'CHR$134;"LOAD A VOCABULARY"
2790 PRINT'"   Load a   previously   created vocabularyfrom ta
pe."
2800 PRINT'"   To exit to the Main Menu   press RETURNin repl
y to the name of the file."
2810 PRINT'"   Put the tape in your cassette recorderready t
o load."
2820 PRINT'"   What is the name of the file":INPUT FILE$
2830 IF FILE$="" THEN GOTO 2930
2840 PRINT'"   Press any key to continue.":Z$=GET$
2850 FILE=OPENIN(FILE$)
2860 INPUT#FILE,LANG$(1),LANG$(2),MAXWD%
2870 FOR I%=1 TO MAXWD%-1
2880   INPUT# FILE,WORD$(I%,0),WORD$(I%,1)

2890 NEXT I%
2900 CLOSE#FILE
2910 PRINT'FILE$;" Loaded OK"
2920 PROC_NEXT
2930 ENDPROC
2940 REM =====
2950 REM   PROC_LIST
2960 REM .....

```

```

2970 REM
2980 DEF PROC_LIST
2990 CLS:PROC_TITLE
3000 PRINT 'CHR$134;"LIST THE VOCABULARY"
3010 IF MAXWD%=1 THEN PRINT'CHR$129;"No words in the vocabu
lary": GOTO 3140
3020 PRINT CHR$130;LANG$(1);TAB(19);CHR$133;LANG$(2):PRINT
3030 J%=1
3040 FOR I%=1 TO MAXWD%-1
3050     PRINT TAB(1);WORD$(I%,0);TAB(20);WORD$(I%,1)
3060     J%=J%+1
3070     IF J%<16 THEN GOTO 3130
3080     PROC_NEXT
3090     CLS:PROC_TITLE
3100     PRINT 'CHR$134;"LIST THE VOCABULARY"
3110     PRINT CHR$130;LANG$(1);TAB(19);CHR$133;LANG$(2):PRIN
T
3120     J%=1
3130 NEXT I%
3140 PROC_NEXT
3150 ENDPROC
3160 REM =====
3170 REM     PROC_DELETE
3180 REM     .....
3190 REM

3200 DEF PROC_DELETE
3210 CLS:PROC_TITLE
3220 PRINT'CHR$134;"DELETE A WORD"
3230 IF MAXWD%=1 THEN PRINT'''CHR$129;"You must enter some
words before you":PRINT CHR$129;"can delete them!!!!!!":PR
OC_NEXT:ENDPROC
3240 PRINT'" You will be asked to give the languageof the
word you wish to delete and then the word."
3250 PRINT'" You will then be asked if you want todelete
another word. This will continueuntil you reply 'N'."
3260 PROC_NEXT
3270 REPEAT
3280     PROC_DELETE_WORD
3290     PROC_YORN
3300 UNTIL NO=TRUE
3310 ENDPROC
3320 REM =====
3330 REM PROC_DELETE_WORD
3340 REM     .....
3350 REM
3360 DEF PROC_DELETE_WORD
3370 CLS:PROC_TITLE
3380 PRINT'CHR$134;"DELETE A WORD"
3390 PRINT'" To exit to the Main Menu press RETURNin repl
y to any question."
3400 PRINT'"Type in the language;":INPUT LANG$
3410 IF LANG$="" THEN GOTO 3580
3420 LANG$=FNCONV(LANG$)
3430 IF LANG$<>LANG$(1) AND LANG$<>LANG$(2) THEN PRINT'CHR
$129;"Language unknown!": GOTO 3580
3440 IF LANG$=LANG$(1) THEN LANG%=0 ELSE LANG%=1
3450 PRINT'"Type the word to delete in ";LANG$(LANG%+1):INP
UT WORD$
3460 IF WORD$="" THEN GOTO 3580
3470 WORD$=FNCONV(WORD$)
3480 DUP=FALSE
3490 FOR I%=1 TO MAXWD%-1

3500     IF WORD$=WORD$(I%,LANG%) THEN DUP=TRUE: J%=I%:I%=MAX
WD%-1
3510 NEXT I%
3520 IF NOT DUP THEN PRINT'CHR$129;"No such word!!!!": GOT

```



```

O 3580
3530 FOR I%=J% TO MAXWD%-1
3540     WORD$(I%,0)=WORD$(I%+1,0)
3550     WORD$(I%,1)=WORD$(I%+1,1)
3560 NEXT I%
3570 MAXWD%=MAXWD%-1
3580 PRINT TAB(0,23);CHR$129;"Another word to delete (Y/N)"
;
3590 ENDPROC
3600 REM =====
3610 REM     PROC_LOOK_UP
3620 REM     .....
3630 REM
3640 DEF PROC_LOOK_UP
3650 CLS:PROC_TITLE
3660 IF MAXWD%=1 THEN PRINT'''CHR$129;"You must enter some
words before you":PRINT CHR$129;"can look them up!!!!!!":PR
OC_NEXT:ENDPROC
3670 PRINT 'CHR$134;"LOOK UP A WORD"
3680 PRINT'" You will be asked to give the languageof the
word you wish to look up and then the word. The word will t
hen be shown in the second language and you will be asked i
f you want another word."
3690 PRINT'" This will continue until you reply 'N'"
3700 PROC_NEXT
3710 REPEAT
3720     PROC_LOOK_UP_WORD
3730     PROC_YORN
3740 UNTIL NO=TRUE
3750 ENDPROC
3760 REM =====
3770 REM     PROC_LOOK_UP_WORD
3780 REM     .....
3790 REM
3800 DEF PROC_LOOK_UP_WORD
3810 CLS:PROC_TITLE
3820 PRINT'CHR$134;"LOOK UP A WORD"
3830 PRINT'" To exit to the Main Menu press RETURN in repl
y to any question."
3840 PRINT'"Type the language":INPUT LANG$
3850 IF LANG$="" THEN GOTO 3990
3860 LANG$=FNCONV(LANG$)
3870 IF LANG$<>LANG$(1) AND LANG$<>LANG$(2) THEN PRINT'"CHR
$129;"Language unknown!!!!":GOTO 3990
3880 IF LANG$=LANG$(1) THEN LANG%=0 ELSE LANG%=1
3890 PRINT '"Type the word to look up in ";LANG$(LANG%+1):I
NPUT WORD$
3900 IF WORD$="" THEN GOTO 3990
3910 WORD$=FNCONV(WORD$)
3920 DUP=FALSE
3930 FOR I%=1 TO MAXWD%-1
3940     IF WORD$=WORD$(I%,LANG%) THEN DUP=TRUE: J%=I%:I%=MAX
WD%-1
3950 NEXT I%
3960 IF NOT DUP THEN PRINT'CHR$129;"No matching word found!
!!!!":GOTO 3990
3970 PRINT'"The word in ";LANG$((LANG%+1) MOD 2+1);" is "
3980 PRINT WORD$(J%,(LANG%+1)MOD 2)
3990 PRINT TAB(0,23);CHR$129;"Another word to look up (Y/N)
";
4000 ENDPROC
4010 REM =====
4020 REM     ERRORS COME HERE
4030 REM     .....
4040 REM
4050 IF ERR=17 THEN RUN
4060 MODE7
4070 REPORT: PRINT" at line ";ERL
4080 END

```

